

Supplementary Information:

Figure S5: NONMEM control stream of the final model

```
$PROBLEM
$INPUT      ID
            TIME ; [h]
            DV   ; [ng/mL]
            MDV
            AMT  ; [mg]
            CMT
            EVID
            L2
            ASSAY ; 1=plasma (EDTA), 3=DBS (automated)
            WT   ; [kg]
$DATA      ../DerivedData/IVM_FINAL.DTA IGNORE=@

$SUBROUTINE ADVAN6 TOL=5
$MODEL      COMP=(ABS) ; 1
            COMP=(CENT) ; 2
            COMP=(PER) ; 3
$PK
  IF (AMT.GT.0) PODO=AMT

  TVCL      = THETA(1) * (WT/70)**0.75
  CL        = TVCL * EXP(ETA(1)) ; L/h
  TVV2      = THETA(2) * WT/70
  V2        = TVV2 * EXP(ETA(2)) ; L
  TVKA      = THETA(3)
  KA        = TVKA ; 1/h
  TVV3      = THETA(4)
  V3        = TVV3 * EXP(ETA(3)) ; L
  TVQ       = THETA(5)
  Q         = TVQ * EXP(ETA(4)) ; L/h
  TVMTT     = THETA(6)
  MTT       = TVMTT * EXP(ETA(5)) ; h
  TVNN      = THETA(7)
  NN        = TVNN

  S2        = V2/1000 ; dose in mg, DV in ng/mL

  F1 = 0

  KEL = CL/V2
  K23 = Q/V2
  K32 = Q/V3
  KTR = (NN+1)/MTT

  A_0(1) = 0
  A_0(2) = 0
  A_0(3) = 0

  LNFAC = 6.579251 ; log(6!)

$DES
  X = 0.00001 ; add in calls to log() to prevent errors from taking log(0)
  DADT(1) = EXP(LOG(PODO+X)+LOG(KTR+X)+NN*LOG(KTR*T+X)-KTR*T-LNFAC) -
KA*A(1)
  DADT(2) = KA*A(1) - KEL*A(2) - K23*A(2) + K32*A(3)
  DADT(3) = K23*A(2) - K32*A(3)

$ERROR
  IPRED = F
  RATIO = THETA(8)
  IF (ASSAY.EQ.3) IPRED=IPRED*RATIO

  W = IPRED
  IF (W.EQ.0) W=1
```

```

W2 = 1

IF (ASSAY.EQ.1) Y = IPRED + W*EPS(1) + EPS(3)
IF (ASSAY.EQ.3) Y = IPRED + W*EPS(2) + EPS(3)

$ESTIMATION METHOD=1 PRINT=5 MAX=9999 SIG=3 INTER
$COVARIANCE

$THETA
(0,7.66829           ; CL
(0,89.1294           ; V2
(0,0.553456          ; KA
(0,234.328           ; V3
(0,18.9994           ; Q
(0.0001,1.05086)     ; MTT
6 FIX                ; N
(0.0001,0.710273)    ; RATIO

$OMEGA
0.0637389           ; IIV_CL
0.0105281           ; IIV_V2
0.042025            ; IIV_V3
0.165925            ; IIV_Q
0.35847             ; IIV_MTT

$SIGMA BLOCK(2)
0.00878848           ; PROP_ERROR_PLASMA
0.00433558 0.0124671 ; PROP_ERROR_DBS
$SIGMA 0.508876      ; ADD_ERROR

```